

Simulace s vícerozměrným regresním modelem

Úloha: Simulace dat pro další použití, zacházení s vícerozměrnými daty.

- dvourozměrný vstup i výstup (lze měnit)
- první řád modelu (lze měnit)

Simulují se data z normálního, dvourozměrného regresního modelu

$$y_t = b_0 u_t + a_1 y_{t-1} + b_1 u_{t-1} + k + e_t$$

kde y , u , e jsou výstup, vstup a porucha, a , b , k jsou (maticové) parametry.

Předpoklady: $e \sim N(0, r)$, r konstantní; vstup je generován předem.

Poznámka

Obecně - y_t má dimenzi ny a u_t má dimenzi nu . Pak a_i jsou matice $ny \times ny$, b_i matice $ny \times nu$, k a e_t jsou sloupcové vektory o dimenzi ny .

Regresní vektor je

$$\psi_t = [u'_t, y'_{t-1}, u'_{t-1} \cdots y'_{t-n}, u'_{t-n}, 1]'$$

je to sloupcový vektor s dimenzí $nf = (n+1) \cdot nu + n \cdot ny + 1$ a

vektor parametrů je

$$\theta = [b_0, a_1, b_1 \cdots a_n, b_n, k]$$

což je matice s dimenzí $ny \times nf$.

Pozor: Ve vícerozměrném případě má součin regresního vektoru a parametru tvar

$$\theta \psi_t$$

a výsledkem je sloupec s dimenzí ny

Značení

- y - y_t ,
- u - u_t ,
- a , b , k - a , b , k , (použití viz níže),
- r - cv (použití viz níže).

Volitelné parametry

- nd - počet simulovaných dat
- Sim.Cy.ord - řád modelu
- I_typU - volba řízení
- Sim.Cy.thb0
Sim.Cy.tha1
Sim.Cy.thb1
Sim.Cy.thk - parametry modelu
(značení: Sim - simulace, Cy - komponenta pro spojitý výstup)

Doporučené experimenty

1. Měňte parametry simulované soustavy a sledujte vstup a výstup soustavy na grafech.
2. Měňte počet vstupů a výstupů soustavy.
Pozor: parametry soustavy jsou nastaveny pro dva vstupy a dva výstupy. Pokud chcete tyto rozměry měnit, musíte dodat do programu vlastní parametry. Ty pak určují počet vstupů a výstupů. Parametry u mnohorozměrné soustavy jsou matice.

Program

```
// Simulation of a two-dimensional first order regression model
// -----
[u,t,n]=file();                // find working directory
chdir(dirname(n(1)));          // set working directory
clear("u","t","n")           // clear auxiliary data
exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=100;                        // length of data
I_typU=2;                      // type of control (see below)

// model for simulation
Sim.Cy.ord=1;                  // model order
Sim.Cy.thb0=[1 .2; -.5 1];     // parameters at ut
Sim.Cy.tha1=[.4 .2; .1 .5];   // parameters at yt
Sim.Cy.thb1=[.1 -2; -.2 .1];
Sim.Cy.thk=[0; 0];            // constant (model absolute term)
Sim.Cy.th=[Sim.Cy.thb0 Sim.Cy.tha1 Sim.Cy.thb1 Sim.Cy.thk];
Sim.Cy.cv=.1;                 // noise variance

[my mu]=size(Sim.Cy.thb0);
yt=zeros(my,nd);
yt(1)=1; yt(2)=3;            // initial conditions for output

// definition of the control variable
select I_typU
```

```

case 1, ut=rand(mu,nd);           // random
case 2
    z=1;
    for i=1:mu
        ut(i,1:nd)=z*ones(1,nd);
        z=-z;
    end
end

// TIME LOOP OF THE SIMULATION
ord=Sim.Cy.ord;                   // model order
sd=sqrt(Sim.Cy.cv);
th=Sim.Cy.th;                    // regression coefficients
for t=3:nd
    ps=genps(ord,t,yt,ut);       // regression vector
    yt(:,t)=th*ps+sd*eye(my)*rand(my,1); // simulation
end
Sim.Cy.yt=yt;
Sim.Cy.ut=ut;

save _data/dataT11N.dat Sim      // save data

// RESULTS OF THE SIMULATION
set(scf(1),'position',[600 100 600 600])
subplot(211),plot(ut','.'),title('Input')
subplot(212),plot(yt'),title('Output')

```